



APIs by Example: Reading an IFS Directory

[System i Network Programming Tips Newsletter](#)

System iNEWS Staff

Thu, 03/15/2001 (All day)

This is the first installment of APIs by Example, which offers sample code for using the most popular OS/400 APIs in RPG IV and CL. We'll run this feature in every issue of the Club Tech Programming Tips newsletter, so be sure to stay tuned for more APIs!

This week's API example shows how to read each entry found of a specified IFS directory. There are four basic steps in the routine:

- a. Open the specified directory. If it fails, return a null value.
- b. Read all directory entries. The end of directory is denoted with a null value.
- c. For each directory entry, convert time values with localtime and ctime, and retrieve user names with getpwuid. If it fails, return a 1.
- d. Close the opened directory. If it fails, return a 1.

Although this is a small routine, it uses five APIs and two C library functions, which are listed below.

Unix-type APIs:

- a. Open directory (opendir)

Opens a directory so that it can be read with the readdir function and returns a pointer to be used by subsequent dir operations. <http://publib.boulder.ibm.com/pubs/html/as400/v4r5/ic2924/info/apis/opendir.htm>

- b. Read directory entry (readdir)

Reads the directory entries one by one, returning a pointer to a dirent structure describing each entry.

<http://publib.boulder.ibm.com/pubs/html/as400/v4r5/ic2924/info/apis/readdir.htm>

- c. Close directory (closedir)

Closes the directory stream and frees the buffer used by readdir.

<http://publib.boulder.ibm.com/pubs/html/as400/v4r5/ic2924/info/apis/closedir.htm>

- d. Get file or link information (lstat)

Gets status information about the specified directory entry and puts it in the structure pointed to by the pBuf parameter. <http://publib.boulder.ibm.com/pubs/html/as400/v4r5/ic2924/info/apis/lstat.htm>

- e. Get user information for user ID (getpwuid)

Returns information describing the user identified by the uid parameter.

<http://publib.boulder.ibm.com/pubs/html/as400/v4r5/ic2924/info/apis/getpwuid.htm>

C library functions:

- a. Convert time (localtime)

Converts a time value to a structure of type tm (see definition in code sample).

<http://publib.boulder.ibm.com/cgi-bin/bookmgr/BOOKS/qb3at500/1.2.84>

- b. Convert time to character string (ctime)

Converts a time value to local time in the form of a character string. <http://publib.boulder.ibm.com/cgi-bin/bookmgr/BOOKS/qb3at500/1.2.27>

<http://publib.boulder.ibm.com/cgi-bin/bookmgr/BOOKS/qb3at500/1.2.27>

Below is the sample RPG IV code.

```

**
** Compile options:
**
** CRTBNDRPG ... DBGVIEW(*LIST)
**
**
**-- Control spec: -----**
H Option( *SrcStmt ) DftActGrp( *No ) BndDir( 'QC2LE' )
**-- Global variables: -----**
D RtnCod s 10i 0
D DirNam s 64a Inz( '/QTCPTMM' )
D Time s 24a
D User s 10a
**-- Dir entry-structure: -----**
D DirEnt Ds Based( pDirEnt ) Align
D d_reserved1 16a
D d_fileno 10u 0
D d_reclen 10u 0
D d_reserved3 10i 0
D d_reserved4 6a
D d_reserved5 2a
D d_nlsinfo
D d_ccsid 10i 0 Overlay( d_nlsinfo: 1 )
D d_country_id 2a Overlay( d_nlsinfo: 5 )
D d_lang_id 3a Overlay( d_nlsinfo: 7 )
D d_nls_reserv 3a Overlay( d_nlsinfo: 10 )
D d_namelen 10u 0
D d_name 640a
**
D pDir s *
**-- File stat-structure: -----**
D Buf Ds Align
D st_mode 10u 0
D st_ino 10u 0
D st_nlink 5u 0
D 2a
D st_uid 10u 0
D st_gid 10u 0
D st_size 10i 0
D st_atime 10i 0
D st_mtime 10i 0
D st_ctime 10i 0
D st_dev 10u 0
D st_blksize 10u 0
D st_alloca 10u 0
D st_objtype 11a
D 1a
D st_codepage 5u 0
D st_reserved1 62a
D st_ino_gen_id 10u 0
**
D pBuf s * Inz( %Addr( Buf ) )
**-- Time tm-structure: -----**
D Tm Ds Based( pTm )
**-- Seconds (0-61)
D tm_sec 10i 0
**-- Minutes (0-59)
D tm_min 10i 0
**-- Hours (0-23)
D tm_hour 10i 0
**-- Day of month (1-31)
D tm_mday 10i 0
**-- Month (0-11)
D tm_mon 10i 0
**-- Year (-1900)
D tm_year 10i 0
**-- Day of week (0-6)
D tm_wday 10i 0
**-- Day of year (0-365)
D tm_yday 10i 0
**-- DST in effect
D tm_isdst 10i 0
**-- Passwd structure: -----**

```

```

D Passwd Ds Based( pPasswd ) Align
D pw_name *
D pw_uid 10u 0
D pw_gid 10u 0
D pw_dir *
D pw_shell *
**-- Directory functions: -----**
D opendir Pr * ExtProc( 'opendir' )
D * Value Options( *String )
**
D readdir Pr * ExtProc( 'readdir' )
D * Value
**
D closedir Pr 10i 0 ExtProc( 'closedir' )
D * Value
**-- File functions: -----**
D lstat Pr 10i 0 ExtProc( 'lstat' )
D * Value Options( *String )
D * Value
**-- Time functions: -----**
D localtime Pr * ExtProc( 'localtime' )
D 10i 0 Const
**
D ctime Pr * ExtProc( 'ctime' )
D 10i 0 Const
**-- User functions: -----**
D getpwuid Pr * ExtProc( 'getpwuid' )
D 10u 0 Value
**
**-- Retrieve directory entry attributes: -----**
**
C Eval pDir = opendir( %TrimR( DirNam ) )
C Eval pDirEnt = readdir( pDir )
**
C DoW pDirEnt <> *Null
**
C If lstat( %TrimR( DirNam ) + '/' +
C %Str( %Addr( d_name ) )
C : pBuf
C ) <> -1
**
**-- Extract time attributes - tm-structure:
C Eval pTm = localtime( st_atime )
C Eval pTm = localtime( st_mtime )
C Eval pTm = localtime( st_ctime )
**
**-- Extract time attributes - formatted string:
C Eval Time = %Str( ctime( st_atime ) )
C Eval Time = %Str( ctime( st_mtime ) )
C Eval Time = %Str( ctime( st_ctime ) )
**
C Eval pPasswd = getpwuid( st_uid )
C Eval User = %Str( pw_name )
**
C EndIf
**
C Eval pDirEnt = readdir( pDir )
C EndDo
**
C Eval RtnCod = closedir( pDir )
**
C Return

```

The above tip was written by Carsten Flensburg. For questions regarding this tip, contact Carsten at flensburg@novasol.dk

Source URL: <http://iprodeveloper.com/development/apis-example-reading-ifs-directory>