



Validate CCSID Utility

[System i Network Systems Management Newsletter](#)

Chuck Lundgren

Wed, 04/14/2004 (All day)

This issue's utility by Carsten Flensburg is command VALCCSID (Validate CCSID), which provides a quick way of determining if your iSeries supports the given Coded Character Set ID (CCSID). The utility consists of command source CBX911X, RPG program CBX911, and the help source CBX911H. The source appears below.

a. Command source CBX911X

```

/*-----*/
/*                                          */
/*  Compile options:                      */
/*                                          */
/*      CrtCmd Cmd( VALCCSID )            */
/*      Pgm( CBX911 )                    */
/*      SrcMbr( CBX911X )                 */
/*      HlpPnlGrp( CBX911H )              */
/*      HlpId( *CMD )                    */
/*                                          */
/*-----*/
Cmd      Prompt( 'Validate CCSID' )

Parm      CCSID          *INT4          +
          Min( 1 )        +
          Range( 1 65535 ) +
          Expr( *YES )     +
          Prompt( 'Coded character set ID' )

```

b. RPG program CBX911

```

**
**
**  Program . . : CBX911
**  Description : Validate CCSID
**  Author . . : Carsten Flensburg
**
**
**  Compilation specification:
**
**      CrtRpgMod Module( CBX911 )
**      DbgView( *NONE )
**
**      CrtPgm      Pgm( CBX911 )
**      Module( CBX911 )
**      ActGrp( QILE )
**
**
**-- Header specifications: -----**
H Option( *SrcStmt )
**-- Send program message: -----**
D SndPgmMsg      Pr          ExtPgm( 'QMHSNDPM' )
D SpMsgId        7a      Const
D SpMsgFq        20a     Const
D SpMsgDta       128a     Const
D SpMsgDtaLen    10i 0 Const
D SpMsgTyp       10a      Const
D SpCalStkE      10a      Const Options( *VarSize )
D SpCalStkCtr    10i 0 Const
D SpMsgKey       4a
D SpError        10i 0 Const

```

```

**-- Run system command: -----**
D ValidateCCSID   Pr           10i 0 ExtProc( 'QtgValidateCCSID'
)

D CCSID           10i 0 Value
**-- Send completion message: -----**
D SndCmpMsg       Pr           10i 0
D PxMsgDta        512a  Const  Varying
**-- Global variable: -----**
D CCSID           s           5s 0
**-- Parameter: -----**
D PxCCSID         s           10i 0
**
C      *Entry      Plist
C                      Parm          PxCCSID
**
**-- Mainline: -----**
**
C                      Eval      CCSID      = PxCCSID
**
C                      If        ValidateCCSID( PxCCSID ) > *Zero
**
C                      CallP      SndCmpMsg( 'CCSID '          +
C                                      %EditC( CCSID: 'X' )    +
C                                      ' is valid.'            +
C                                      )
**
C                      Else
C                      CallP      SndCmpMsg( 'CCSID '          +
C                                      %EditC( CCSID: 'X' )    +
C                                      ' is not supported.'    +
C                                      )
C                      EndIf
**
C                      Return
**
**-- Send completion message: -----**
P SndCmpMsg       B
D                      Pi           10i 0
D PxMsgDta        512a  Const  Varying
**
D MsgKey          s           4a
**
C                      CallP(e) SndPgmMsg( 'CPF9897'
C                                      : 'QCPFMSG *LIBL'
C                                      : PxMsgDta
C                                      : %Len( PxMsgDta )
C                                      : '*COMP'
C                                      : '*PGMBDY'
C                                      : 1
C                                      : MsgKey
C                                      : *Zero
C                                      )
**
C                      If      %Error
C                      Return   -1
**
C                      Else
C                      Return    0
C                      EndIf
**
P SndCmpMsg       E

```

c. Help source CBX911H:

```

.*-----**
.*
.* Compile options:
.*
.*      CrtPnlGrp PnlGrp( CBX911H )
.*              SrcFile( QPNLSRC )
.*              SrcMbr( CBXP11H )
.*
.*-----**

```

```
:PNLGRP.  
:HELP NAME='VALCCSID'.Validate CCSID - Help  
:P.  
The Validate Coded Character Set ID command determines if a given  
CCSID  
is actually supported by the iSeries server performing the validation.  
:P.  
:EHELP.  
:HELP NAME='VALCCSID/CCSID'.Coded character set ID (CCSID) - Help  
:XH3.Coded character set ID (CCSID)  
:P.  
Specifies the coded character set ID to be validated.  
:P.  
This is a required parameter.  
:P.  
:EHELP.  
:EPNLGRP.
```

The above utility was written by Carsten Flensburg. For questions regarding this tip, contact Carsten at <mailto:flensburg@novasol.dk>.

Source URL: <http://iprodeveloper.com/systems-management/validate-ccsid-utility>